
SZÁMÍTÓGÉPI GRAFIKA 
VÁGÁS 

FELADAT: 

Ha az alakzat nagyobb, mint a képtartomány , amelyben megjelenítendő, akkor a kívül eső része-

ket el kell hagyni, azaz az alakzatról „le kell vágni”, röviden szólva: az alakzatot vágni kell. 

Tegyük föl, hogy az alakzat (egyenes) szakaszokra bontható. (A tartományok vágása hasonlóan 

megoldható, bár további problémákat is fölvet.) Az alábbiakban szakaszok vágását vizsgáljuk. 

Cohen-Sutherland módszere 

Cohen-Sutherland algoritmusa a következő felismerésre épít: 

A szakasz és a képtartomány kölcsönös viszonya lényegében a következő négyféle lehet: a sza-

kasz 

1. teljesen benne van a képtartományban (l. az ábrán az A szakaszt), 

2. csak egyik végpontja esik a tartományba (l. az ábrán az B szakaszt), 

3. bár mindkét végpontja kívül esik a tartományon mégis a „középső” darabja a tartományban 

fut (l. az ábrán az C szakaszt), 

4. és a tartomány nem rendelkezik közös ponttal (l. az ábrán az D szakaszt). 

 

1. ábra. Szakasz és képtartomány viszonya. 

Mindenek előtt próbálja ki: mire lehet gondolni (csvagasFP.exe,/CSvagasTP.exe)!  

                                                 
 Szokás szerint a képernyővel megegyező állású téglalapra gondolunk képtartomány esetén. 

csvagasFP.exe
CSvagasTP.exe

A lényeg 

Osszuk fel a képtartományt magába foglaló síkot 9 részre, és sorszámozzuk meg az alábbiak 

szerint. Vegyük észre a sorszámokban a „bináris szabályszerűséget”!  

9=10012 1=00012 3=00112 

8=10002 0=00002 2=00102 

12=11002 4=01002 6=01102 

(-: Ugye megvan a bináris logika? :-) 

Egy-egy bit tartozik –a felső, a „balsó”, az alsó és a „jobbsó”– a négy kívül levő síkrészhez.  

A vágandó szakasz végpontjait kódoljuk a szerint, melyik tartományba esik. Ha a szakasz mind-

két végpontja  0-kódú, akkor kirajzolható. Ha nem, akkor gondolkodni kell! :-) 

Vegyük észre: akkor és csak akkor fut a szakasz „egyértelműen szerencsétlen” tartomány(ok)ba, 

ha a két végpont-kód bites ÉS-művelet eredménye nem 0 értékű. (Például: 9 ÉS 3 = 10012 ÉS 

00112 = 00012… éppen az a bit 1-es, amelyik a „szerencsétlen” tartományhoz van rendelve.) 

Ha egyéb eset áll fenn, akkor egyszer vagy kétszer metszi a (0-kódú) képtartomány határoló 

egyeneseit. A feladat e metszéspontok megtalálása. Valahogy így : 

Eljárás CSVagas(Változó p1,p2:TPont): 

    Változó  

      c,c1,c2:TKod;  p:TPont 

    c1:=VégpontKód(p1); c2:=VégpontKód(p2) 

    Ciklus amíg nem ((c1=0) és (c2=0)) [legalább egyik végpont még nincs belül] 

                és ((c1 ÉS c2)=0)  [a szakasz nem biztos, hogy kívül fut]  

      Ha c1=0 akkor [p1 OK. Nézzük a másikat: Csere(p1,p2)] 

        p:=p1; p1:=p2; p2:=p;  c:=c1; c1:=c2; c2:=c 

      Elágazás vége 

      [p1 felöli új metszéspont meghatározása => p1] 

      Elágazás 

         (c1 ÉS 1)=1 esetén FelsoVagas(p1.x,p1.y,p2.x,p2.y) 

         (c1 ÉS 2)=2 esetén JobbVagas(p1.x,p1.y,p2.x,p2.y) 

         (c1 ÉS 4)=4 esetén AlsoVagas(p1.x,p1.y,p2.x,p2.y) 

         (c1 ÉS 8)=8 esetén BalVagas(p1.x,p1.y,p2.x,p2.y) 

      Elágazás vége 

      c1:=VégpontKód(p1) 

    Ciklus vége 

                                                 
 balsó  1xxx2, alsó  x1xx2, jobbsó  xx1x2, felső  xxx12 

 A logikai műveletek jelei kisbetűsek, a bit-műveletek NAGYBETŰSEK. 

    Ha (c1=0) és (c2=0) akkor {a belső szakasz kirajzolható} 

      Szakasz(p1,p2) 

    Elágazás vége 

Eljárás vége. 

 

2. ábra. Képernyő-koordinátarendszer és képtartomány 

Függvény VégpontKód(Konstans p:TPont):TKód 

  Változó 

    c:TKod 

  c:=0 

  Ha p.y<KF akkor c:=c VAGY 1 

  Ha p.x>KJ akkor c:=c VAGY 2 

  Ha p.y>KA akkor c:=c VAGY 4 

  Ha p.x<KB akkor c:=c VAGY 8 

  VégpontKód:=c 

Eljárás vége. 

Az egyes vágáseljárások meghatározzák a p1 és p2 pontra illeszkedő egyenes és a megfelelő 

határoló egyenes metszéspontját. Ez két lineáris egyenletből álló egyenletrendszer megoldását 

jelenti. 

Például a FelsőVágás esetében: az y=(y2-y1)/(x2-x1)*(x-x1)+y1 és az y=KF egyenletek megoldá-

sát keressük. Azaz x-re kell rendezni a KF=(y2-y1)/(x2-x1)*(x-x1)+y1 egyenletet, amiből már 

következik a keresett x és y:  

x=x1+(x2-x1)/(y2-y1)*(KF-y1) 

y=KF 

Azaz az eljárás kódja: 

Eljárás FelsoVagas(Változó x1,y1,x2,y2:Egész
1
): 

  x1:=x1+(x2-x1)/(y2-y1)*(KF-y1); y1:=KF 

Eljárás vége. 

A program az alábbi deklarációk után válik teljesen érthetővé, egyértelművé: 

Típus  TPont=Rekord(x,y:Egész) 

       TKod=0..12 

                                                 
1
 Világos, hogy valójában csak az 1. pont koordinátái fognak változni, így precízebb lenne az alábbi szignatúra: 

 Eljárás FeloVagas(Változó x1,y1:Egész, Konstans x2,y2:Egész) 



 4 

 

3. ábra. A Cohen-Sutherland algoritmusa működés közben. 

A (Turbo/Free Pascal) keretprogram olvasható alább, amelyben csak néhány eljárás törzse maradt 

kitöltendő. (Letölthető a Free Pascal-hoz: LetoltendoVagasFP.zip) 

A KERETPROGRAM 

Program Vagas; 1 
  Uses 2 
    Crt,Graph; 3 
  Const 4 
    gPath='C:\langs\tp\bgi'; 5 
    {KépTartományt határoló egyenesek paraméterei (VGA-hoz igazítva):} 6 
    KF0=100; KB0=100; KJ0=500; KA0=350; 7 
    MaxX0=640; MaxY0=480;     {Próbaszakaszok száma:} 8 
    SzakaszDb=5; 9 
  Type 10 
    TPont=Record x,y:Integer End; 11 
    TSzakasz=Array [1..2] of TPont; 12 
    TKod=0..12; 13 
  Const {Próbaszakaszok:} 14 
    sz:Array [1..SzakaszDb] of TSzakasz=( 15 
                                 ((x:KB0+229;y:KF0+129),(x:KJ0-229;y:KA0-129)), 16 
                                 ((x:KB0-19;y:KF0+19),(x:KB0+190;y:KF0+39)), 17 
                                 ((x:KB0-9;y:KA0-19),(x:KJ0+19;y:KA0-99)), 18 
                                 ((x:KB0-9;y:KF0-9),(x:KJ0+19;y:KF0-19)), 19 
                                 ((x:KB0-59;y:KA0+59),(x:KJ0+69;y:KF0-69)) 20 
                                ); 21 
  Var 22 
    p1,p2:TPont; 23 
    i:Integer; 24 
    KF,KB,KA,KJ:Integer; 25 
 26 

LetoltendoVagasFP.zip

  Procedure Inic; 27 
    Var 28 
      i,j, 29 
      gd,gm:Integer; 30 
      nyX,nyY:Real; 31 
  Begin 32 
    DetectGraph(gd,gm); 33 
    InitGraph(gd,gm,gPath); 34 
    ClearDevice; 35 
    {képernyőre transzformálás (TP->FP miatt):} 36 
    nyX:=GetMaxX/MaxX0; nyY:=GetMaxY/MaxY0; 37 
    KF:=Round(KF0*nyY); KA:=Round(KA0*nyY); 38 
    KB:=Round(KB0*nyX); KJ:=Round(KJ0*nyX); 39 
    For i:=1 to SzakaszDb do 40 
      For j:=1 to 2 do 41 
      Begin 42 
        sz[i][j].x:=Round(sz[i][j].x*nyX); sz[i][j].y:=Round(sz[i][j].y*nyY); 43 
      End; 44 
    SetLineStyle(SolidLn, 0, ThickWidth); 45 
    SetColor(Green); Rectangle(KB,KF,KJ,KA); 46 
  End; 47 
 48 
  Procedure Szakasz(Const p1,p2:TPont); 49 
  Begin 50 
    Line(p1.x,p1.y,p2.x,p2.y); 51 
  End; 52 
 53 
  Function VegpontKod(Const p:TPont):TKod; 54 
  Begin 55 
  56 
  End; 57 
 58 
  Procedure FelsoVagas(Var x1,y1,x2,y2:Integer); 59 
  Begin 60 
  61 
  End; 62 
 63 
  Procedure AlsoVagas(Var x1,y1,x2,y2:Integer); 64 
  Begin 65 
  66 
  End; 67 
 68 
  Procedure BalVagas(Var x1,y1,x2,y2:Integer); 69 
  Begin 70 
  71 
  End; 72 
 73 
  Procedure JobbVagas(Var x1,y1,x2,y2:Integer); 74 
  Begin 75 
  76 
  End; 77 

 78 
  Procedure CSVagas(Var p1,p2:TPont); 79 
    Var 80 
      c,c1,c2:TKod; 81 
      p:TPont; 82 
  Begin 83 
  84 
  End; 85 
 86 
Begin 87 
  Inic; 88 
  For i:=1 to SzakaszDb do 89 
  Begin 90 
    SetColor(Red); p1:=sz[i][1]; p2:=sz[i][2]; 91 
    Szakasz(p1,p2); 92 
    ReadKey; 93 
    CSVagas(p1,p2); 94 
    ReadKey 95 
  End; 96 
End. 97 

Minden egyben letölthető: Vagas.zip. 

A MEGOLDÁS: 

 
Program Vagas; 1 
  Uses 2 
    Crt,Graph; 3 
  Const 4 
    gPath='C:\langs\tp\bgi'; 5 
    {KépTartományt határoló egyenesek paraméterei (VGA-hoz igazítva):} 6 
    KF0=100; KB0=100; KJ0=500; KA0=350; 7 
    MaxX0=640; MaxY0=480;  8 
    {Próbaszakaszok száma:} 9 
    SzakaszDb=5; 10 
  Type 11 
    TPont=Record x,y:Integer End; 12 
    TSzakasz=Array [1..2] of TPont; 13 
    TKod=0..12; 14 
  Const 15 
    sz:Array [1..SzakaszDb] of TSzakasz=( 16 
                                 ((x:KB0+229;y:KF0+129),(x:KJ0-229;y:KA0-129)), 17 
                                 ((x:KB0-19;y:KF0+19),(x:KB0+190;y:KF0+39)), 18 
                                 ((x:KB0-9;y:KA0-19),(x:KJ0+19;y:KA0-99)), 19 
                                 ((x:KB0-9;y:KF0-9),(x:KJ0+19;y:KF0-19)), 20 
                                 ((x:KB0-59;y:KA0+59),(x:KJ0+69;y:KF0-69)) 21 
                                 ); 22 
  Var 23 
    p1,p2:TPont; 24 
    i:Integer; 25 
    KF,KB,KA,KJ:Integer; 26 
 27 
  Procedure Inic; 28 
    Var 29 
      i,j, 30 
      gd,gm:Integer; 31 
      nyX,nyY:Real;  32 
  Begin 33 
    DetectGraph(gd,gm); 34 
    InitGraph(gd,gm,gPath); 35 
    ClearDevice; 36 
    {képernyőre transzformálás (TP->FP miatt):} 37 
    nyX:=GetMaxX/MaxX0; nyY:=GetMaxY/MaxY0; 38 
    KF:=Round(KF0*nyY); KA:=Round(KA0*nyY); 39 
    KB:=Round(KB0*nyX); KJ:=Round(KJ0*nyX); 40 
    For i:=1 to SzakaszDb do 41 
      For j:=1 to 2 do 42 
      Begin 43 
        sz[i,j].x:=Round(sz[i,j].x*nyX); sz[i,j].y:=Round(sz[i,j].y*nyY); 44 
      End; 45 

    SetLineStyle(SolidLn, 0, ThickWidth); 46 
    SetColor(Green); Rectangle(KB,KF,KJ,KA); 47 
  End; 48 
 49 
  Procedure Szakasz(Const p1,p2:TPont); 50 
  Begin 51 
    Line(p1.x,p1.y,p2.x,p2.y); 52 
  End; 53 
 54 
  Function VegpontKod(Const p:TPont):TKod; 55 
  Var 56 
    c:TKod; 57 
  Begin 58 
    c:=0; 59 
    If p.y<KF then c:=c OR 1; 60 
    If p.x>KJ then c:=c OR 2; 61 
    If p.y>KA then c:=c OR 4; 62 
    If p.x<KB then c:=c OR 8; 63 
    VegpontKod:=c 64 
  End; 65 
 66 
  Procedure FelsoVagas(Var x1,y1,x2,y2:Integer); 67 
  Begin 68 
    x1:=Round(x1+(x2-x1)/(y2-y1)*(KF-y1));  69 
    y1:=KF 70 
  End; 71 
 72 
  Procedure AlsoVagas(Var x1,y1,x2,y2:Integer); 73 
  Begin 74 
    x1:=Round(x1+(x2-x1)/(y2-y1)*(KA-y1)); 75 
    y1:=KA 76 
  End; 77 
 78 
  Procedure BalVagas(Var x1,y1,x2,y2:Integer); 79 
  Begin 80 
    y1:=Round(y1+(y2-y1)/(x2-x1)*(KB-x1)); 81 
    x1:=KB 82 
  End; 83 
 84 
  Procedure JobbVagas(Var x1,y1,x2,y2:Integer); 85 
  Begin 86 
    y1:=Round(y1+(y2-y1)/(x2-x1)*(KJ-x1)); 87 
    x1:=KJ 88 
  End; 89 
 90 
  Procedure CSVagas(Var p1,p2:TPont); 91 
    Var 92 
      c,c1,c2:TKod; 93 
      p:TPont; 94 

                                                 
 Hogy ne legyen a részletszámítások során sem túlcsordulás az első adandó helyen osztunk. 

  Begin 95 
    VegpontKod(c1,p1); VegpontKod(c2,p2); 96 
    While not ((c1=0) and (c2=0)) {legalább egyik még nincs belül} 97 
          and ((c1 AND c2)=0)  {nem biztos, hogy kívül} do 98 
    Begin 99 
      If c1=0 then {p1 OK. Nézzük a másikat: Csere(p1,p2)} 100 
      Begin 101 
        c:=c1; c1:=c2; c2:=c; 102 
        p:=p1; p1:=p2; p2:=p; 103 
      End; 104 
      {p1 felöli új metszéspont meghatározása => p1} 105 
      If 106 
        (c1 AND 1)=1 then FelsoVagas(p1.x,p1.y,p2.x,p2.y) else if 107 
        (c1 AND 2)=2 then JobbVagas(p1.x,p1.y,p2.x,p2.y) else if 108 
        (c1 AND 4)=4 then AlsoVagas(p1.x,p1.y,p2.x,p2.y) else if 109 
        (c1 AND 8)=8 then BalVagas(p1.x,p1.y,p2.x,p2.y) 110 
      {EndIf}; 111 
      VegpontKod(c1,p1); 112 
      {Metszéspont-keresés -- nyomkövetés:} 113 
      SetLineStyle(CenterLn, 0, NormWidth); 114 
      SetColor(Yellow); 115 
      Szakasz(p1,p2); 116 
      ReadKey; 117 
      SetLineStyle(SolidLn, 0, ThickWidth); 118 
      {Nyomkövetés vége} 119 
    End; 120 
    If (c1=0) and (c2=0) then 121 
    Begin {a belső szakasz kirajzolható} 122 
      SetColor(Blue); 123 
      Szakasz(p1,p2); 124 
    End; 125 
  End; 126 
 127 
Begin 128 
  Inic; 129 
  For i:=1 to SzakaszDb do 130 
  Begin 131 
    SetColor(Red); p1:=sz[i][1]; p2:=sz[i][2]; 132 
    Szakasz(p1,p2); 133 
    ReadKey; 134 
    CSVagas(p1,p2); 135 
    ReadKey 136 
  End; 137 
End. 138 


